
Subject: Re: Fonts in Z-device

Posted by [davidf](#) on Mon, 07 Feb 2000 08:00:00 GMT

[View Forum Message](#) <> [Reply to Message](#)

Andy Loughe (loughe@fsl.noaa.gov) writes:

> Nice (hardware) fonts in 'X'... no problem!
> Nice (hardware) fonts in 'PS'... no problem!
> Nice (hardware) fonts in 'Z'... now there's a problem!
>
> I thought the Z-device would know about the fonts
> available in 'X', but that is not the case. I
> even tried device, set_font = 'Times' and although
> I did not get an error, I got vector fonts!
>
> What's a Z-device user to do!?

I like to use true-type fonts, but you have to use
the ol' "make-um big" trick to make them look really
nice. Here is an example:

PRO ZBuffer_Test

data = Dist(41)

Window, XSize=400, YSize=400, Title='Normal Picture'

Surface, data, XTitle='This is X Title', YTitle='This is Y Title', \$

 ZTitle='This is Z Title', Font=1, CharSize=2.5, \$

 Position=[0.15, 0.15, 0.90, 0.90, 0.15, 0.90]

thisDevice = !D.Name

Set_Plot, 'Z'

Device, Set_Resolution=[1600, 1600]

Surface, data, XTitle='This is X Title', YTitle='This is Y Title', \$

 ZTitle='This is Z Title', Font=1, CharSize=2.5*4, \$

 XThick=4, YThick=4, ZThick=4, Thick=4, \$

 Position=[0.15, 0.15, 0.90, 0.90, 0.15, 0.90]

snapshot = TVRD()

Set_Plot, thisDevice

Window, 1, XSize=400, YSize=400, Title='Z Buffer Picture'

TV, Rebin(snapshot, 400, 400)

END

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting

Phone: 970-221-0438 E-Mail: davidf@dfanning.com
Coyote's Guide to IDL Programming: <http://www.dfanning.com/>
Toll-Free IDL Book Orders: 1-888-461-0155
